
NAG Library Routine Document

G05ZQF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

G05ZQF performs the setup required in order to simulate stationary Gaussian random fields in two
dimensions, for a user-defined variogram, using the circulant embedding method. Specifically, the
eigenvalues of the extended covariance matrix (or embedding matrix) are calculated, and their square
roots output, for use by G05ZSF, which simulates the random field.

2 Specification

SUBROUTINE G05ZQF (NS, XMIN, XMAX, YMIN, YMAX, MAXM, VAR, COV2, EVEN,
PAD, ICORR, LAM, XX, YY, M, APPROX, RHO, ICOUNT, EIG,
IUSER, RUSER, IFAIL)

&
&

INTEGER NS(2), MAXM(2), EVEN, PAD, ICORR, M(2), APPROX,
ICOUNT, IUSER(*), IFAIL

&

REAL (KIND=nag_wp) XMIN, XMAX, YMIN, YMAX, VAR, LAM(MAXM(1)*MAXM(2)),
XX(NS(1)), YY(NS(2)), RHO, EIG(3), RUSER(*)

&

EXTERNAL COV2

3 Description

A two-dimensional random field Z xð Þ in R
2 is a function which is random at every point x 2 R

2, so
Z xð Þ is a random variable for each x. The random field has a mean function � xð Þ ¼ E Z xð Þ½ � and a
symmetric positive semidefinite covariance function C x; yð Þ ¼ E Z xð Þ � � xð Þð Þ Z yð Þ � � yð Þð Þ½ �. Z xð Þ is
a Gaussian random field if for any choice of n 2 N and x1; . . . ; xn 2 R

2, the random vector
Z x1ð Þ; . . . ; Z xnð Þ½ �T follows a multivariate Normal distribution, which would have a mean vector ~�� with
entries ~�i ¼ � xið Þ and a covariance matrix ~C with entries ~Cij ¼ C xi; xj

� �
. A Gaussian random field

Z xð Þ is stationary if � xð Þ is constant for all x 2 R
2 and C x; yð Þ ¼ C xþ a; yþ að Þ for all x; y; a 2 R

2

and hence we can express the covariance function C x; yð Þ as a function � of one variable:
C x; yð Þ ¼ � x� yð Þ. � is known as a variogram (or more correctly, a semivariogram) and includes the
multiplicative factor �2 representing the variance such that � 0ð Þ ¼ �2.

The routines G05ZQF and G05ZSF are used to simulate a two-dimensional stationary Gaussian random
field, with mean function zero and variogram � xð Þ, over a domain xmin ; xmax½ � � ymin ; ymax½ �, using an
equally spaced set of N1 �N2 points; N1 points in the x-direction and N2 points in the y-direction. The
problem reduces to sampling a Normal random vector X of size N1 �N2, with mean vector zero and a
symmetric covariance matrix A, which is an N2 by N2 block Toeplitz matrix with Toeplitz blocks of
size N1 by N1. Since A is in general expensive to factorize, a technique known as the circulant
embedding method is used. A is embedded into a larger, symmetric matrix B, which is an M2 by M2

block circulant matrix with circulant blocks of size M1 by M1, where M1 � 2 N1 � 1ð Þ and
M2 � 2 N2 � 1ð Þ. B can now be factorized as B ¼ W�W � ¼ R�R, where W is the two-dimensional
Fourier matrix (W � is the complex conjugate of W ), � is the diagonal matrix containing the

eigenvalues of B and R ¼ �
1
2W � . B is known as the embedding matrix. The eigenvalues can be

calculated by performing a discrete Fourier transform of the first row (or column) of B and multiplying
by M1 �M2, and so only the first row (or column) of B is needed – the whole matrix does not need to
be formed.

The symmetry of A as a block matrix, and the symmetry of each block of A, depends on whether the

variogram � is even or not. � is even in its first coordinate if � �x1; x2½ �T
� �

¼ � x1; x2½ �T
� �

, even in its

second coordinate if � x1;�x2½ �T
� �

¼ � x1; x2½ �T
� �

, and even if it is even in both coordinates (in two

dimensions it is impossible for � to be even in one coordinate and uneven in the other). If � is even
then A is a symmetric block matrix and has symmetric blocks; if � is uneven then A is not a symmetric
block matrix and has non-symmetric blocks. In the uneven case, M1 and M2 are set to be odd in order
to guarantee symmetry in B.

As long as all of the values of � are non-negative (i.e., B is positive semidefinite), B is a covariance
matrix for a random vector Y which has M2 blocks of size M1. Two samples of Y can now be
simulated from the real and imaginary parts of R� Uþ iVð Þ, where U and V have elements from the

standard Normal distribution. Since R� Uþ iVð Þ ¼ W�
1
2 Uþ iVð Þ , this calculation can be done using a

discrete Fourier transform of the vector �
1
2 Uþ iVð Þ . Two samples of the random vector X can now be

recovered by taking the first N1 elements of the first N2 blocks of each sample of Y – because the
original covariance matrix A is embedded in B, X will have the correct distribution.

If B is not positive semidefinite, larger embedding matrices B can be tried; however if the size of the
matrix would have to be larger than MAXM, an approximation procedure is used. We write
� ¼ �þ þ ��, where �þ and �� contain the non-negative and negative eigenvalues of B respectively.
Then B is replaced by �Bþ where Bþ ¼ W�þW � and � 2 0; 1ð � is a scaling factor. The error � in
approximating the distribution of the random field is given by

� ¼
ffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffi
1� �ð Þ2 trace�þ �2 trace��

M

s
:

Three choices for � are available, and are determined by the input argument ICORR:

setting ICORR ¼ 0 sets

� ¼ trace�

trace�þ
;

setting ICORR ¼ 1 sets

� ¼
ffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffi
trace�

trace�þ

s
;

setting ICORR ¼ 2 sets � ¼ 1.

G05ZQF finds a suitable positive semidefinite embedding matrix B and outputs its sizes in the vector M
and the square roots of its eigenvalues in LAM. If approximation is used, information regarding the
accuracy of the approximation is output. Note that only the first row (or column) of B is actually
formed and stored.

4 References

Dietrich C R and Newsam G N (1997) Fast and exact simulation of stationary Gaussian processes
through circulant embedding of the covariance matrix SIAM J. Sci. Comput. 18 1088–1107

Schlather M (1999) Introduction to positive definite functions and to unconditional simulation of
random fields Technical Report ST 99–10 Lancaster University

Wood A T A and Chan G (1994) Simulation of stationary Gaussian processes in 0; 1½ �d Journal of
Computational and Graphical Statistics 3(4) 409–432

5 Arguments

1: NSð2Þ – INTEGER array Input

On entry: the number of sample points to use in each direction, with NSð1Þ sample points in the
x-direction, N1 and NSð2Þ sample points in the y-direction, N2. The total number of sample
points on the grid is therefore NSð1Þ � NSð2Þ.

Constraints:

NSð1Þ � 1;
NSð2Þ � 1.

2: XMIN – REAL (KIND=nag_wp) Input

On entry: the lower bound for the x-coordinate, for the region in which the random field is to be
simulated.

Constraint: XMIN < XMAX.

3: XMAX – REAL (KIND=nag_wp) Input

On entry: the upper bound for the x-coordinate, for the region in which the random field is to be
simulated.

Constraint: XMIN < XMAX.

4: YMIN – REAL (KIND=nag_wp) Input

On entry: the lower bound for the y-coordinate, for the region in which the random field is to be
simulated.

Constraint: YMIN < YMAX.

5: YMAX – REAL (KIND=nag_wp) Input

On entry: the upper bound for the y-coordinate, for the region in which the random field is to be
simulated.

Constraint: YMIN < YMAX.

6: MAXMð2Þ – INTEGER array Input

On entry: determines the maximum size of the circulant matrix to use – a maximum of
MAXMð1Þ elements in the x-direction, and a maximum of MAXMð2Þ elements in the
y-direction. The maximum size of the circulant matrix is thus MAXMð1Þ�MAXMð2Þ.
Constraints:

i f EVEN ¼ 1, MAXMðiÞ � 2k, where k is the smallest integer sat isfying
2k � 2 NSðiÞ � 1ð Þ, for i ¼ 1; 2;
if EVEN ¼ 0, MAXMðiÞ � 3k, where k is the smallest integer sat isfying
3k � 2 NSðiÞ � 1ð Þ, for i ¼ 1; 2.

7: VAR – REAL (KIND=nag_wp) Input

On entry: the multiplicative factor �2 of the variogram � xð Þ.
Constraint: VAR � 0:0.

8: COV2 – SUBROUTINE, supplied by the user. External Procedure

COV2 must evaluate the variogram � xð Þ for all x if EVEN ¼ 0, and for all x with non-negative
entries if EVEN ¼ 1. The value returned in GAMMA is multiplied internally by VAR.

The specification of COV2 is:

SUBROUTINE COV2 (X, Y, GAMMA, IUSER, RUSER)

INTEGER IUSER(*)
REAL (KIND=nag_wp) X, Y, GAMMA, RUSER(*)

1: X – REAL (KIND=nag_wp) Input

On entry: the coordinate x at which the variogram � xð Þ is to be evaluated.

2: Y – REAL (KIND=nag_wp) Input

On entry: the coordinate y at which the variogram � xð Þ is to be evaluated.

3: GAMMA – REAL (KIND=nag_wp) Output

On exit: the value of the variogram � xð Þ.

4: IUSERð�Þ – INTEGER array User Workspace
5: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

COV2 is called with the arguments IUSER and RUSER as supplied to G05ZQF. You
should use the arrays IUSER and RUSER to supply information to COV2.

COV2 must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which G05ZQF is called. Arguments denoted as Input must not be changed by this
procedure.

9: EVEN – INTEGER Input

On entry: indicates whether the covariance function supplied is even or uneven.

EVEN ¼ 0
The covariance function is uneven.

EVEN ¼ 1
The covariance function is even.

Constraint: EVEN ¼ 0 or 1.

10: PAD – INTEGER Input

On entry: determines whether the embedding matrix is padded with zeros, or padded with values
of the variogram. The choice of padding may affect how big the embedding matrix must be in
order to be positive semidefinite.

PAD ¼ 0
The embedding matrix is padded with zeros.

PAD ¼ 1
The embedding matrix is padded with values of the variogram.

Suggested value: PAD ¼ 1.

Constraint: PAD ¼ 0 or 1.

11: ICORR – INTEGER Input

On entry: determines which approximation to implement if required, as described in Section 3.

Suggested value: ICORR ¼ 0.

Constraint: ICORR ¼ 0, 1 or 2.

12: LAMðMAXMð1Þ �MAXMð2ÞÞ – REAL (KIND=nag_wp) array Output

On exit: contains the square roots of the eigenvalues of the embedding matrix.

13: XXðNSð1ÞÞ – REAL (KIND=nag_wp) array Output

On exit: the points of the x-coordinates at which values of the random field will be output.

14: YYðNSð2ÞÞ – REAL (KIND=nag_wp) array Output

On exit: the points of the y-coordinates at which values of the random field will be output.

15: Mð2Þ – INTEGER array Output

On exit: Mð1Þ contains M1, the size of the circulant blocks and Mð2Þ contains M2, the number of
blocks, resulting in a final square matrix of size M1 �M2.

16: APPROX – INTEGER Output

On exit: indicates whether approximation was used.

APPROX ¼ 0
No approximation was used.

APPROX ¼ 1
Approximation was used.

17: RHO – REAL (KIND=nag_wp) Output

On exit: indicates the scaling of the covariance matrix. RHO ¼ 1:0 unless approximation was
used with ICORR ¼ 0 or 1.

18: ICOUNT – INTEGER Output

On exit: indicates the number of negative eigenvalues in the embedding matrix which have had to
be set to zero.

19: EIGð3Þ – REAL (KIND=nag_wp) array Output

On exit: indicates information about the negative eigenvalues in the embedding matrix which
have had to be set to zero. EIGð1Þ contains the smallest eigenvalue, EIGð2Þ contains the sum of
the squares of the negative eigenvalues, and EIGð3Þ contains the sum of the absolute values of
the negative eigenvalues.

20: IUSERð�Þ – INTEGER array User Workspace
21: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by G05ZQF, but are passed directly to COV2 and should be
used to pass information to this routine.

22: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, NS ¼ valueh i; valueh i½ �.
Constraint: NSð1Þ � 1, NSð2Þ � 1.

IFAIL ¼ 2

On entry, XMIN ¼ valueh i and XMAX ¼ valueh i.
Constraint: XMIN < XMAX.

IFAIL ¼ 4

On entry, YMIN ¼ valueh i and YMAX ¼ valueh i.
Constraint: YMIN < YMAX.

IFAIL ¼ 6

On entry, MAXM ¼ valueh i; valueh i½ �.
Constraint: the minima for MAXM are valueh i; valueh i½ �.
Where, if EVEN ¼ 1, the minimum calculated value of MAXMðiÞ is given by 2k, where k is the
smallest integer satisfying 2k � 2 NSðiÞ � 1ð Þ, and if EVEN ¼ 0, the minimum calculated value
of MAXMðiÞ is given by 3k, where k is the smallest integer satisfying 3k � 2 NSðiÞ � 1ð Þ, for
i ¼ 1; 2.

IFAIL ¼ 7

On entry, VAR ¼ valueh i.
Constraint: VAR � 0:0.

IFAIL ¼ 9

On entry, EVEN ¼ valueh i.
Constraint: EVEN ¼ 0 or 1.

IFAIL ¼ 10

On entry, PAD ¼ valueh i.
Constraint: PAD ¼ 0 or 1.

IFAIL ¼ 11

On entry, ICORR ¼ valueh i.
Constraint: ICORR ¼ 0, 1 or 2.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

If on exit APPROX ¼ 1, see the comments in Section 3 regarding the quality of approximation;
increase the values in MAXM to attempt to avoid approximation.

8 Parallelism and Performance

G05ZQF is threaded by NAG for parallel execution in multithreaded implementations of the NAG
Library.

G05ZQF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor
library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example calls G05ZQF to calculate the eigenvalues of the embedding matrix for 25 sample points
on a 5 by 5 grid of a two-dimensional random field characterized by the symmetric stable variogram:

� xð Þ ¼ �2 exp � x0ð Þ�� �
;

where x0 ¼ x
‘1
þ y

‘2

��� ��� , and ‘1, ‘2 and � are parameters.

It should be noted that the symmetric stable variogram is one of the pre-defined variograms available in
G05ZRF. It is used here purely for illustrative purposes.

10.1 Program Text

! G05ZQF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

Module g05zqfe_mod

! G05ZQF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: cov2

! .. Parameters ..
Integer, Parameter, Public :: even = 1

Contains
Subroutine cov2(t1,t2,gamma,iuser,ruser)

! .. Implicit None Statement ..
Implicit None

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out) :: gamma
Real (Kind=nag_wp), Intent (In) :: t1, t2

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
Integer, Intent (Inout) :: iuser(*)

! .. Local Scalars ..
Real (Kind=nag_wp) :: l1, l2, nu, rnorm, tl1, tl2
Integer :: norm

! .. Intrinsic Procedures ..
Intrinsic :: abs, exp, sqrt

! .. Executable Statements ..

! Covariance parameters stored in ruser array.
norm = iuser(1)
l1 = ruser(1)
l2 = ruser(2)
nu = ruser(3)

tl1 = abs(t1)/l1
tl2 = abs(t2)/l2
If (norm==1) Then

rnorm = tl1 + tl2
Else If (norm==2) Then

rnorm = sqrt(tl1**2+tl2**2)
End If

gamma = exp(-(rnorm**nu))

Return

End Subroutine cov2
End Module g05zqfe_mod

Program g05zqfe

! G05ZQF Example Main Program

! .. Use Statements ..
Use nag_library, Only: g05zqf, nag_wp
Use g05zqfe_mod, Only: cov2, even

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: l1, l2, nu, rho, var, xmax, xmin, &

ymax, ymin
Integer :: approx, icorr, icount, ifail, norm, &

pad
! .. Local Arrays ..

Real (Kind=nag_wp) :: eig(3), ruser(3)
Real (Kind=nag_wp), Allocatable :: lam(:), xx(:), yy(:)
Integer :: iuser(1), m(2), maxm(2), ns(2)

! .. Executable Statements ..
Write (nout,*) ’G05ZQF Example Program Results’
Write (nout,*)

! Get problem specifications from data file
Call read_input_data(norm,l1,l2,nu,var,xmin,xmax,ymin,ymax,ns,maxm, &

icorr,pad)

! Put covariance parameters in communication arrays
iuser(1) = norm
ruser(1) = l1
ruser(2) = l2
ruser(3) = nu

Allocate (lam(maxm(1)*maxm(2)),xx(ns(1)),yy(ns(2)))

! Get square roots of the eigenvalues of the embedding matrix
ifail = 0
Call g05zqf(ns,xmin,xmax,ymin,ymax,maxm,var,cov2,even,pad,icorr,lam,xx, &

yy,m,approx,rho,icount,eig,iuser,ruser,ifail)

! Output results
Call display_results(approx,m,rho,eig,icount,lam)

Contains
Subroutine read_input_data(norm,l1,l2,nu,var,xmin,xmax,ymin,ymax,ns, &

maxm,icorr,pad)

! .. Implicit None Statement ..
Implicit None

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (Out) :: l1, l2, nu, var, xmax, xmin, ymax, &

ymin
Integer, Intent (Out) :: icorr, norm, pad

! .. Array Arguments ..
Integer, Intent (Out) :: maxm(2), ns(2)

! .. Executable Statements ..
! Skip heading in data file

Read (nin,*)

! Read in norm, l1, l2 and nu for cov2 function
Read (nin,*) norm, l1, l2, nu

! Read in variance of random field
Read (nin,*) var

! Read in domain endpoints
Read (nin,*) xmin, xmax
Read (nin,*) ymin, ymax

! Read in number of sample points in each direction
Read (nin,*) ns(1), ns(2)

! Read in maximum size of embedding matrix
Read (nin,*) maxm(1), maxm(2)

! Read in choice of scaling in case of approximation
Read (nin,*) icorr

! Read in choice of padding
Read (nin,*) pad

Return

End Subroutine read_input_data

Subroutine display_results(approx,m,rho,eig,icount,lam)

! .. Implicit None Statement ..
Implicit None

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: rho
Integer, Intent (In) :: approx, icount

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: eig(3)
Integer, Intent (In) :: m(2)
Real (Kind=nag_wp), Intent (In) :: lam(m(1),m(2))

! .. Local Scalars ..
Integer :: i

! .. Executable Statements ..
! Display size of embedding matrix

Write (nout,*)
Write (nout,99999) ’Size of embedding matrix = ’, m(1)*m(2)

! Display approximation information if approximation used
Write (nout,*)
If (approx==1) Then

Write (nout,*) ’Approximation required’
Write (nout,*)
Write (nout,99998) ’RHO = ’, rho
Write (nout,99997) ’EIG = ’, eig(1:3)
Write (nout,99999) ’ICOUNT = ’, icount

Else
Write (nout,*) ’Approximation not required’

End If

! Display square roots of the eigenvalues of the embedding matrix
Write (nout,*)
Write (nout,*) ’Square roots of eigenvalues of embedding matrix:’
Write (nout,*)
Do i = 1, m(1)

Write (nout,99996) lam(i,1:m(2))
End Do

Return

99999 Format (1X,A,I7)
99998 Format (1X,A,F10.5)
99997 Format (1X,A,3(F10.5,1X))
99996 Format (1X,8F8.4)

End Subroutine display_results

End Program g05zqfe

10.2 Program Data

G05ZQF Example Program Data
2 0.1 0.15 1.2 : norm, l1, l2, nu
0.5 : var

-1 1 : xmin, xmax
-0.5 0.5 : ymin, ymax
5 5 : ns

81 81 : maxm
2 : icorr
1 : pad

10.3 Program Results

G05ZQF Example Program Results

Size of embedding matrix = 64

Approximation not required

Square roots of eigenvalues of embedding matrix:

0.8966 0.8234 0.6810 0.5757 0.5391 0.5757 0.6810 0.8234
0.8940 0.8217 0.6804 0.5756 0.5391 0.5756 0.6804 0.8217
0.8877 0.8175 0.6792 0.5754 0.5391 0.5754 0.6792 0.8175
0.8813 0.8133 0.6780 0.5751 0.5390 0.5751 0.6780 0.8133
0.8787 0.8116 0.6774 0.5750 0.5390 0.5750 0.6774 0.8116
0.8813 0.8133 0.6780 0.5751 0.5390 0.5751 0.6780 0.8133
0.8877 0.8175 0.6792 0.5754 0.5391 0.5754 0.6792 0.8175
0.8940 0.8217 0.6804 0.5756 0.5391 0.5756 0.6804 0.8217

